4　数式の計算

Mathematica は数値だけではなく，多項式，有理式，また $\sqrt{x}$，$\sin x$，$\log x$ などを含む無理式も扱える．

■ 多項式

多項式の計算方法について，解説する．まず，変数 f と g とに多項式を割り当てる．

```
In[1]:= f = x + 2 y - 2 + x y

Out[1]= -2 + x + 2 y + x y

In[2]:= g = x^2 + 3 x y + y - 2 y - 1
```

Out[2]= $-1 + x^2 - y + 3\,x\,y$

表示された多項式の項の順序が入力時と異なることに注意． Mathematica は多項式の項を昇ベキの順で並び替える．また，同類項は一つにまとめられる．

■ 和と積

この二つの多項式の和をとる．同類項の計算および昇ベキの順の並び替えは自動的に行われる．

```
In[3]:= f + g
```

Out[3]= $-3 + x + x^2 + y + 4\,x\,y$

次に fとg との積をとる．積は自動的には展開されない．展開をするには Expand を用いる．また，展開された多項式を因数分解するには Factor を用いる．

```
In[4]:= f g
```

Out[4]= $(-2 + x + 2\,y + x\,y)\,(-1 + x^2 - y + 3\,x\,y)$

```
In[5]:= Expand[%]
```

Out[5]= $2 - x - 2\,x^2 + x^3 - 8\,x\,y + 5\,x^2\,y + x^3\,y - 2\,y^2 + 5\,x\,y^2 + 3\,x^2\,y^2$

```
In[6]:= Factor[%]
```

Out[6]= $(-2 + x + 2\,y + x\,y)\,(-1 + x^2 - y + 3\,x\,y)$

ベキ乗は ^ を用いる．

```
In[7]:= f^3
```

Out[7]= $(-2 + x + 2\,y + x\,y)^3$

■ 項をまとめる

指定された変数に関して同じ次数の項をまとめるには， `Collect` を用いる．先ほどの式を， `x` に関して同じ次数の項をまとめてみる．

In[8]:= **Collect[f g, x]**

Out[8]= $2-2y^2+x^3(1+y)+x^2(-2+5y+3y^2)+x(-1-8y+5y^2)$

次に， `y` に関してまとめてみる．

In[9]:= **Collect[f g, y]**

Out[9]= $2-x-2x^2+x^3+(-8x+5x^2+x^3)y+(-2+5x+3x^2)y^2$

■ 係数

多項式中の指定された項の係数を取り出すには， `Coefficient` を用いる．

In[10]:= **Coefficient[f g, x^2]**

Out[10]= $-2+5y+3y^2$

指定された変数に対する各次の係数を取り出すには， `CoefficientList` を用いる．結果は0次から昇ベキの順のリストとなる．

In[11]:= **CoefficientList[f g, x]**

Out[11]= $\{2-2y^2, -1-8y+5y^2, -2+5y+3y^2, 1+y\}$

■ 項，因数の取り出し

多項式の完全形式を `FullForm` で見てみる．

In[12]:= **g**

Out[12]= $-1+x^2-y+3xy$

In[13]:= **FullForm[g]**

Out[13]//FullForm=

```
Plus[-1, Power[x, 2], Times[-1, y], Times[3, x, y]]
```

多項式全体の頭部は `Plus` で，一つひとつの項がその要素になっている．この中から，たとえば3 x yを取り出すには，要素の番号を `[[ ]]` の中で指定すればよい．

In[14]:= **g[[4]]**

Out[14]= $3xy$

`Collect` と組み合わせれば，特定の変数に関して特定の次数の項だけを抜き出すことができる．

```
In[15]:= Collect[f g, x]
```

Out[15]= $2-2y^2+x^3(1+y)+x^2(-2+5y+3y^2)+x(-1-8y+5y^2)$

```
In[16]:= %[[5]]
```

Out[16]= $x(-1-8y+5y^2)$

また，因数分解をした形の頭部は `Times` になっており，因数がその要素になっている．

```
In[17]:= h = Factor[(x^3 + y^3)]
```

Out[17]= $(x+y)(x^2-xy+y^2)$

```
In[18]:= FullForm[h]
```

```
Out[18]//FullForm=
      Times[Plus[x, y], Plus[Power[x, 2], Times[-1, x, y], Power[y, 2]]]
```

したがって，`[[ ]]` を用いると指定した因数を取り出すことができる．

```
In[19]:= h[[2]]
```

Out[19]= x^2-xy+y^2

練習問題： $x^{100}-1$ を因数分解せよ．その中で最大の次数の因数を取り出せ．

練習問題： `Expand` を用いて $(x+y+z)^{10}$ の $x^2y^3z^5$ の係数を求めよ．それが $\frac{10!}{2!\,3!\,5!}$ であることを確認せよ．また，$(x+y+z)^{10}$ の展開式の中で `x` の次数が `2` 次である項を集めた式を求めよ．

■ 有理式

有理式（分数式）も，そのままの形で入力できる．ただし，必要な場合には分子分母を括弧でくくるのを忘れないように．

```
In[20]:= h = (x + 5) / (x + 3) - 6 / (x + 4)
```

Out[20]= $-\frac{6}{4+x}+\frac{5+x}{3+x}$

■ 通分，分子と分母の取り出しおよび展開

有理式の分母を通分するには `Together` を用いる．

```
In[21]:= k = Together[h]
```

Out[21]= $\frac{2+3x+x^2}{(3+x)(4+x)}$

有理式の分子，分母はそれぞれ `Numerator,Denominator` を用いると取り出すことができる．

```
In[22]:= Numerator[k]
```

Out[22]= $2 + 3x + x^2$

```
In[23]:= Denominator[k]
```

Out[23]= $(3 + x)(4 + x)$

分子あるいは分母を展開するには， ExpandNumerator あるいは ExpandDenominator を用いる．

```
In[24]:= ExpandDenominator[k]
```

Out[24]= $\frac{2 + 3x + x^2}{12 + 7x + x^2}$

■ 因数分解，約分，部分分数展開

分子分母を因数分解するには， Factor を用いる．

```
In[25]:= Factor[h]
```

Out[25]= $\frac{(1 + x)(2 + x)}{(3 + x)(4 + x)}$

分子分母を約分するには， Cancel を用いる．

```
In[26]:= Cancel[(x^2 - 1) / (x^3 - 1)]
```

Out[26]= $\frac{1 + x}{1 + x + x^2}$

有理式を，簡単な分子分母をもつ有理式の和に分解する（部分分数展開）には， Apart を用いる．

```
In[27]:= Apart[h]
```

Out[27]= $1 + \frac{2}{3 + x} - \frac{6}{4 + x}$

練習問題： $\frac{x+y}{x+1} - \frac{x}{y+2} + \frac{y}{x+2}$ を通分せよ．

練習問題： $\frac{1}{x^8-1}$ を部分分数展開せよ．その項のなかで，分母の次数が最も高いものを[[ ]]を用いて取り出せ．

■ 無理式

■ 根号

n乗根は，$\frac{1}{n}$ 乗のことなので， ^ すなわち Power を用いれば入力できる．

```
In[28]:= (x + 2 y) ^ (1 / 3)
```

Out[28]= $(x + 2y)^{1/3}$

In[29]:= **Power[x + 2 y, 1 / 3]**

Out[29]= $(x+2y)^{1/3}$

平方根は Sqrt を用いても入力できる.

In[30]:= **Sqrt[x]**

Out[30]= $\sqrt{x}$

指数法則を用いたベキ乗の展開 $(a\,b)^n \to a^n\,b^n$ は PowerExpand を用いる.

In[31]:= **f = (a b) ^ (1 / 2)**

Out[31]= $\sqrt{a\,b}$

In[32]:= **g = PowerExpand[f]**

Out[32]= $\sqrt{a}\,\sqrt{b}$

この式変形が正しいのは a, b 共に正のときであることに注意して, PowerExpand を用いるべきである. たとえば a, b に負の値をいれると, 次のように x, y の値が異なることになる.

In[33]:= **f /. {a -> -2, b -> -3}**

Out[33]= $\sqrt{6}$

In[34]:= **g /. {a -> -2, b -> -3}**

Out[34]= $-\sqrt{6}$

ただし, n が整数のときには PowerExpand の式変形は a, b の正負にかかわらず常に正しい. それゆえ, n が整数のときには, Mathematica は PowerExpand を用いなくとも自動的に展開した形に変形する.

In[35]:= **(a b) ^ 3**

Out[35]= $a^3\,b^3$

■ 指数関数, 対数関数

指数関数 e^x は Exp[x] で入力できるが, その完全形式は Power[E, x] である.

In[36]:= **Exp[x]**

Out[36]= e^x

In[37]:= **FullForm[%]**

Out[37]//FullForm=

```
Power[E, x]
```

$e^a\,e^b$は自動的に e^{a+b} に変形される.

In[38]:= **Exp[a] Exp[b]**

Out[38]= e^{a+b}

自然対数 $\log x = \log_e x$ はLog で表される.

In[39]:= **Log[x]**

Out[39]= Log[x]

In[40]:= **Log[E^3]**

Out[40]= 3

b を底とした対数 $\log_b x$ は Log[b, x] で表される.

In[41]:= **Log[10, 1000]**

Out[41]= 3

展開公式 $\log a\,b = \log a + \log b$ の左辺を右辺に変形するには, PowerExpand を用いる.

In[42]:= **PowerExpand[Log[a b / c]]**

Out[42]= Log[a] + Log[b] - Log[c]

■ 三角関数

通常の三角関数の頭文字を大文字にした, Sin, Cos, Tan などが, Mathematica での関数名である. 三角関数に関しては, いろいろな公式があり, それらを用いて, 種々の変形ができる.

TrigExpand は式を単純な三角関数の積の和に展開する.

In[43]:= **TrigExpand[Sin[3 x] Cos[2 x]]**

Out[43]= $\frac{\mathrm{Sin}[x]}{2} + \frac{5}{2}\mathrm{Cos}[x]^4\,\mathrm{Sin}[x] - 5\,\mathrm{Cos}[x]^2\,\mathrm{Sin}[x]^3 + \frac{\mathrm{Sin}[x]^5}{2}$

TrigFactor は式を三角関数の式の積の形に因数分解する.

In[44]:= **TrigFactor[%]**

Out[44]= (1 + 2 Cos[2 x]) (Cos[x] - Sin[x]) Sin[x] (Cos[x] + Sin[x])

TrigRreduce は整数倍の角度を用いて簡単な式に変形する.

In[45]:= **TrigReduce[%]**

Out[45]= $\frac{1}{2}(\mathrm{Sin}[x] + \mathrm{Sin}[5\,x])$

指数関数と三角関数との間には, $e^{i\,x} = \cos x + i \sin x$ という関係がある. これを用いて指数関数を三角関数で表す変形, またその逆の変形ができる. これには ExpToTrig と TrigToExp とを用いる.

In[46]:= **ExpToTrig[E^(I x)]**

Out[46]= Cos[x] + i Sin[x]

In[47]:= **TrigToExp[Sin[x]]**

Out[47]= $\frac{1}{2}\,i\,e^{-i\,x} - \frac{1}{2}\,i\,e^{i\,x}$

練習問題： 三角関数 sin, cos, tan の２倍角，３倍角の公式を Mathematica を用いて求めよ.

■ 式の簡約化

複雑な式を簡約化するには Simplify または FullSimplify を用いる． Simplify は展開，因数分解などのさまざまな変形を試み，最も簡単な式になったものを返す． FullSimplify は Simplify よりも多種類の変形を，式のあらゆる部分に施して，最も簡単な式を返す.

```
In[48]:= Simplify[Cos[x]^2 - 2 Cos[x] Sin[x] - Sin[x]^2]

Out[48]= Cos[2 x] - Sin[2 x]
```

より高度の変形が必要な式は Simplify では簡約化できないこともある.

```
In[49]:= Simplify[ Sec[b x + Log[x]]
           (Cos[b x] - 2 a x Cos[b x + Log[x]] + Cos[b x + 2 Log[x]] - 2 Sin[b x + Log[x]])]

Out[49]= Sec[b x + Log[x]]
           (Cos[b x] - 2 a x Cos[b x + Log[x]] + Cos[b x + 2 Log[x]] - 2 Sin[b x + Log[x]])
```

このような場合でも， FullSimplify を用いれば簡約化できることがある.

```
In[50]:= FullSimplify[%]

Out[50]= -2 (a x - Cos[Log[x]] + Tan[b x + Log[x]])
```

FullSimplify は Simplify よりも強力であるが，計算時間が非常にかかる場合がある．とりあえず Simplify で試してみて，それでもだめなら FullSimplify を用いるべきである.

■ 演習問題

[4-1] $(x+3y+2)^3\ (x^2+xy-4x)^2$ の展開式の項のなかで x の次数が 3 であるものを集めた式を求めよ.

[4-2] x^n-1 を n = 1, 2, 3, 4, 5, 6, 7 について因数分解してみよ．その結果のなかで，因数の数がもっとも多いのはどれか.

[4-3] $\dfrac{x^3+6x^2-7x-4}{2x^4-6x^3+10x^2-18x+12}$ を部分分数展開し，分母の次数が最大である項を取り出せ.

[4-4] sin(2x), sin(3x), …, sin(6x) を sin x と cos x とで表す，２倍角，３倍角，…，６倍角の公式を求めよ.

[4-5] $\dfrac{\sin x+\cos x}{\sin x-\cos x}$ を指数関数で表せ．このとき，できるだけ簡単な式にせよ.

■ この章で出てきた関数

`Expand[expr]`	式 `expr` を展開する
`Factor[expr]`	式 `expr` を因数分解する
`Collect[expr, x]`	式 `expr` の変数 `x` に関して同次の項をまとめる.
`Together[expr]`	有理式 `expr` の分母を通分する.
`Numerator[expr]`	有理式 `expr` の分子を返す.
`Denominator[expr]`	有理式 `expr` の分母を返す.
`ExpandNumerator[expr]`	有理式 `expr` の分子を展開する.
`ExpandDenominator[expr]`	有理式 `expr` の分母を展開する.
`Apart[expr]`	有理式 `expr` を簡単な分子分母をもつ有理式の和に分解する.
`PowerExpand[expr]`	式 `expr` の指数部分を，指数法則を用いて展開する.
`TrigExpand[expr]`	式 `expr` を，単純な三角関数の積の和に展開する.
`TrigFactor[expr]`	式 `expr` を，三角関数の式の積の形に因数分解する.
`TrigReduce[expr]`	式 `expr` を，整数倍の角度を用いて簡単な式に変形する.
`ExpToTrig[expr]`	式 `expr` の中の指数関数を三角関数で表す.
`TrigToExp[expr]`	式 `expr` の中の三角関数を指数関数で表す.
`Simplify[expr]`	式 `expr` を簡単な形に変形する.
`FullSimplify[expr]`	`Simplify` よりも高度な変形を用いて， `expr` を簡単な式に変形する.